/***************************************************************************
 *   oboe.resources.FileRef                                                *
 *   Copyright (C) 2011 by Felipe Manga                                    *
 *   lufeboma@gmail.com                                                    *
 *                                                                         *
 *   Permission is hereby granted, free of charge, to any person obtaining *
 *   a copy of this software and associated documentation files (the       *
 *   "Software"), to deal in the Software without restriction, including   *
 *   without limitation the rights to use, copy, modify, merge, publish,   *
 *   distribute, sublicense, and/or sell copies of the Software, and to    *
 *   permit persons to whom the Software is furnished to do so, subject to *
 *   the following conditions:                                             *
 *                                                                         *
 *   The above copyright notice and this permission notice shall be        *
 *   included in all copies or substantial portions of the Software.       *
 *                                                                         *
 *   THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND,       *
 *   EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF    *
 *   MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. *
 *   IN NO EVENT SHALL THE AUTHORS BE LIABLE FOR ANY CLAIM, DAMAGES OR     *
 *   OTHER LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, *
 *   ARISING FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR *
 *   OTHER DEALINGS IN THE SOFTWARE.                                       *
 ***************************************************************************/

package oboe.resources
{
	import flash.system.ApplicationDomain;
	import flash.utils.Dictionary;
	import flash.utils.getQualifiedClassName;
	import oboe.core.Oboe;
	
	
	public class FileRef {
		public var handle : String;
		public var object : Object;
		public var url : Object;
		public var appDomain : ApplicationDomain;
		public static var onLoadCallback : Function;
		public var staticInstance : Dictionary;
		public var error:Boolean;
		
		// optionally specify a handle -a name withwhich to identify this asset- or use the filename -without path/extensions- instead. 
		public function FileRef( file : Object, handle : String = "" ){
			if( file is String )
			{
				this.url = String(file);
			}
			else
			{
				handle = Class(file.classRef).toString();
				var underscore:int = handle.indexOf("_")+1;
				handle = handle.substr( underscore, handle.length-underscore-1);
				this.url = file;
			}
			if( handle == "" ){
				if( url.substr(0, 6).toLowerCase() == "alias:" )
				{
					handle = url.substr( 6, url.indexOf("@")-6 );
				}
				else
				{
					var begin : int = url.lastIndexOf("/")+1;
					var end : int = url.lastIndexOf(".");
					if( end == -1 ) end = int.MAX_VALUE;
					handle = url.substr(begin, end-begin);
				}
			}
			this.handle = handle;
			this.object = null;
			this.appDomain = null;
		}
		
		public function load() : void {
			if( url is String )
			{
				Manager.getURL( String(url), onLoad);
			}
			else
			{
				Manager.getFromClass( Class(url.classRef), String(url.ext), onLoad);
			}
		}
		
		private function onLoad( e : Object, err:Boolean=false ) : void {
			object = e.object;
			appDomain = e.appDomain;
			this.error = err;
			if( !err || (err && Oboe.flags["IGNORE_LOADER_ERRORS"]) )
			{
				onLoadCallback();
			}
		}
	}

}